Solutions for *Mathematica* Exercises (© P. Fischer)

Exercise 1: Numbers

```
Sqrt[2] (* By default, Mathematica stays with exact expressions *)
```

$\sqrt{2}$

```
N[%, 100] (* Force a numberical evaluation of the previous (%) line *)
```

1.414213562373095048801688724209698078569671875376948073176679737990732478462107038850387534327641573

$$\frac{20!}{45}$$

54 064 489 070 592 000

$\frac{20!}{46}$ `∈ Integers (* We can check whether the result is an integer *)`

False

```
FactorInteger[20!] (* FactorInteger[...]
 generates a list of prime factors with multiplicities *)
```

{{2, 18}, {3, 8}, {5, 4}, {7, 2}, {11, 1}, {13, 1}, {17, 1}, {19, 1}}

```
Sqrt[-4]
```

2 ⅈ

$F1 = \frac{4}{5}$; $F2 = \frac{5}{8}$;

```
F1 + F2
```

$$\frac{57}{40}$$

```
F1 * F2
```

$\frac{1}{2}$

```
1 / 2 - 1 / 3
```

$\frac{1}{6}$

```
Sqrt[F1]
```

$\frac{2}{\sqrt{5}}$

$1 - \frac{\sqrt{2\pi\,100}\left(\frac{100}{e}\right)^{100}}{100\,!}$ // N

0.000832983

$1 - \frac{\sqrt{2\pi\,n}\left(\frac{n}{e}\right)^{n}}{n\,!}$ /. n → 100 // N (* more general, replacing n by a fixed value *)

0.000832983

Exercise 2: Lists and Expressions

alist = Table[n^2, {n, 1, 10}]

```
{1, 4, 9, 16, 25, 36, 49, 64, 81, 100}
```

```
{First[alist], alist[[3]], Last[alist]}
```

```
{1, 9, 100}
```

```
v1 = {1, 2, 3}; v2 = {2, 3, 4};
```

```
v1.v2
```

```
20
```

```
A = {{1, 2}, {3, 4}}
```

```
{{1, 2}, {3, 4}}
```

```
A // MatrixForm
```

$$\begin{pmatrix} 1 & 2 \\ 3 & 4 \end{pmatrix}$$

```
B = Inverse[A]
```

$$\left\{\{-2, 1\}, \left\{\frac{3}{2}, -\frac{1}{2}\right\}\right\}$$

```
A.B // MatrixForm
```

$$\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$$

```
A = 3 x + 5; B = 5 x^2 - 7;
```

```
F = A B // Expand
```

$$-35 - 21 x + 25 x^2 + 15 x^3$$

```
F // Factor
```

$$(5 + 3 x) \left(-7 + 5 x^2\right)$$

```
F / A
```

$$\frac{-35 - 21 x + 25 x^2 + 15 x^3}{5 + 3 x}$$

```
% // Simplify
```

$$-7 + 5 x^2$$

Exercise 3: Manipulations & Replacements

```
Clear[A];
```

```
A = (1 + 3 x) (1 + 6 x)
```

$$(1 + 3 x) (1 + 6 x)$$

```
A // Simplify
```

$$(1 + 3 x) (1 + 6 x)$$

```
A // Expand
```

$1 + 9x + 18x^2$

```
% /. x → 1 + √y
```

$1 + 9\left(1 + \sqrt{y}\right) + 18\left(1 + \sqrt{y}\right)^2$

```
% // Simplify
```

$28 + 45\sqrt{y} + 18y$

```
% // Factor
```

$\left(4 + 3\sqrt{y}\right)\left(7 + 6\sqrt{y}\right)$

```
B = % /. y → -1
```

$10 + 45\,i$

```
Solve[A == B, x]
```

$\left\{\left\{x \to -\frac{3}{2} - i\right\}, \{x \to 1 + i\}\right\}$

Exercise 4: Simple Plotting

```
f1[k_, x_] := Sin[k x];
```

```
Plot[f1[1, x], {x, 0, 2 π}, Frame → True, ImageSize → 200, Filling → Axis]
```



```
Plot[f1[k, x] /. k → {1, 2}, {x, 0, 2 π},
 Frame → True, ImageSize → 200, Filling → Axis]
```




```
Plot[{Sin[x], Cos[x]}, {x, 0, 4 π}, Frame → True, ImageSize → 200, Filling → Axis]
```



Exercise 5: Quadratic Equation & more

Part 1

```
EQ = a x^2 + b x + c == 0
  (* The equation. Make sure all symbols are unused (blue!) *)
```

$c + b\,x + a\,x^2 == 0$

```
Solutions = Solve[EQ, x]
 (* This gives a list of solutions as replacement rules *)
```

$$\left\{\left\{x \to \frac{-b - \sqrt{b^2 - 4\,a\,c}}{2\,a}\right\}, \left\{x \to \frac{-b + \sqrt{b^2 - 4\,a\,c}}{2\,a}\right\}\right\}$$

```
EQ /. Solutions (* Check that the solution indeed satisfy EQ *)
```

$$\left\{c + \frac{b\left(-b - \sqrt{b^2 - 4\,a\,c}\right)}{2\,a} + \frac{\left(-b - \sqrt{b^2 - 4\,a\,c}\right)^2}{4\,a} == 0,\; c + \frac{b\left(-b + \sqrt{b^2 - 4\,a\,c}\right)}{2\,a} + \frac{\left(-b + \sqrt{b^2 - 4\,a\,c}\right)^2}{4\,a} == 0\right\}$$

```
% // Simplify
```

{True, True}

```
x1 = x /. First[Solutions]
```

$$\frac{-b-\sqrt{b^2-4\,a\,c}}{2\,a}$$

```
x2 = x /. Last[Solutions]
```

$$\frac{-b+\sqrt{b^2-4\,a\,c}}{2\,a}$$

```
EQ /. x → x1 // Simplify
```

True

```
EQ /. x → x2 // Simplify
```

True

Part 2

$$EQ3 = a_3\,x^3 + a_2\,x^2 + a_1\,x + a_0 == 0;$$

```
Solve[EQ3, x] // Simplify
```

$$\Bigg\{\Bigg\{x \to -\frac{a_2}{3\,a_3} - \left(2^{1/3}\left(-a_2^2+3\,a_1\,a_3\right)\right) \Big/ \left(3\,a_3\left(-2\,a_2^3+9\,a_1\,a_2\,a_3-27\,a_0\,a_3^2+\sqrt{\left(-4\left(a_2^2-3\,a_1\,a_3\right)^3+\left(2\,a_2^3-9\,a_1\,a_2\,a_3+27\,a_0\,a_3^2\right)^2\right)}\right)^{1/3}\right) + \left(-2\,a_2^3+9\,a_1\,a_2\,a_3-27\,a_0\,a_3^2+\sqrt{\left(-4\left(a_2^2-3\,a_1\,a_3\right)^3+\left(2\,a_2^3-9\,a_1\,a_2\,a_3+27\,a_0\,a_3^2\right)^2\right)}\right)^{1/3} \Big/ \left(3\times 2^{1/3}\,a_3\right)\Bigg\},$$

$$\Bigg\{x \to -\frac{a_2}{3\,a_3} + \left(\left(1+i\sqrt{3}\right)\left(-a_2^2+3\,a_1\,a_3\right)\right) \Big/ \left(3\times 2^{2/3}\,a_3\left(-2\,a_2^3+9\,a_1\,a_2\,a_3-27\,a_0\,a_3^2+\sqrt{\left(-4\left(a_2^2-3\,a_1\,a_3\right)^3+\left(2\,a_2^3-9\,a_1\,a_2\,a_3+27\,a_0\,a_3^2\right)^2\right)}\right)^{1/3}\right) - \left(\left(1-i\sqrt{3}\right)\left(-2\,a_2^3+9\,a_1\,a_2\,a_3-27\,a_0\,a_3^2+\sqrt{\left(-4\left(a_2^2-3\,a_1\,a_3\right)^3+\left(2\,a_2^3-9\,a_1\,a_2\,a_3+27\,a_0\,a_3^2\right)^2\right)}\right)^{1/3}\right) \Big/ \left(6\times 2^{1/3}\,a_3\right)\Bigg\},$$

$$\Bigg\{x \to -\frac{a_2}{3\,a_3} + \left(\left(1-i\sqrt{3}\right)\left(-a_2^2+3\,a_1\,a_3\right)\right) \Big/ \left(3\times 2^{2/3}\,a_3\left(-2\,a_2^3+9\,a_1\,a_2\,a_3-27\,a_0\,a_3^2+\sqrt{\left(-4\left(a_2^2-3\,a_1\,a_3\right)^3+\left(2\,a_2^3-9\,a_1\,a_2\,a_3+27\,a_0\,a_3^2\right)^2\right)}\right)^{1/3}\right) - \left(\left(1+i\sqrt{3}\right)\left(-2\,a_2^3+9\,a_1\,a_2\,a_3-27\,a_0\,a_3^2+\sqrt{\left(-4\left(a_2^2-3\,a_1\,a_3\right)^3+\left(2\,a_2^3-9\,a_1\,a_2\,a_3+27\,a_0\,a_3^2\right)^2\right)}\right)^{1/3}\right) \Big/ \left(6\times 2^{1/3}\,a_3\right)\Bigg\}\Bigg\}$$

```
EQ4 = a4 x^4 + a3 x^3 + a2 x^2 + a1 x + a0 == 0;
```

```
Reduce[EQ4, x] // Simplify
```

$$\left(a_4 == 0 \;\&\&\; \left(\left(a_3 \neq 0 \;\&\&\; \left(x == \text{Root}\left[a_0 + \#1\, a_1 + \#1^2\, a_2 + \#1^3\, a_3 \;\&,\, 1\right] \;||\; x == \text{Root}\left[a_0 + \#1\, a_1 + \#1^2\, a_2 + \#1^3\, a_3 \;\&,\, 2\right] \;||\; x == \text{Root}\left[a_0 + \#1\, a_1 + \#1^2\, a_2 + \#1^3\, a_3 \;\&,\, 3\right]\right)\right) \;||\; \left(\left(x == \frac{-a_1 + \sqrt{a_1^2 - 4\, a_0\, a_2}}{2\, a_2} \;||\; a_2 == 0 \;||\; x + \frac{a_1 + \sqrt{a_1^2 - 4\, a_0\, a_2}}{2\, a_2} == 0\right) \;\&\&\; (a_0 == 0 \;||\; a_1 \neq 0 \;||\; a_2 \neq 0) \;\&\&\; \left(x + \frac{a_0}{a_1} == 0 \;||\; a_1 == 0 \;||\; a_2 \neq 0\right) \;\&\&\; a_3 == 0\right)\right)\right) \;||$$

$$\left(a_4 \neq 0 \;\&\&\; \left(x == \text{Root}\left[a_0 + \#1\, a_1 + \#1^2\, a_2 + \#1^3\, a_3 + \#1^4\, a_4 \;\&,\, 1\right] \;||\; x == \text{Root}\left[a_0 + \#1\, a_1 + \#1^2\, a_2 + \#1^3\, a_3 + \#1^4\, a_4 \;\&,\, 2\right] \;||\; x == \text{Root}\left[a_0 + \#1\, a_1 + \#1^2\, a_2 + \#1^3\, a_3 + \#1^4\, a_4 \;\&,\, 3\right] \;||\; x == \text{Root}\left[a_0 + \#1\, a_1 + \#1^2\, a_2 + \#1^3\, a_3 + \#1^4\, a_4 \;\&,\, 4\right]\right)\right)$$

Part 3

```
Table[1 + x^n, {n, 1, 10}]
```

$\{1 + x,\; 1 + x^2,\; 1 + x^3,\; 1 + x^4,\; 1 + x^5,\; 1 + x^6,\; 1 + x^7,\; 1 + x^8,\; 1 + x^9,\; 1 + x^{10}\}$

```
% // Factor
```

$\{1 + x,\; 1 + x^2,\; (1 + x)\,(1 - x + x^2),\; 1 + x^4,\; (1 + x)\,(1 - x + x^2 - x^3 + x^4),\; (1 + x^2)\,(1 - x^2 + x^4),\; (1 + x)\,(1 - x + x^2 - x^3 + x^4 - x^5 + x^6),\; 1 + x^8,\; (1 + x)\,(1 - x + x^2)\,(1 - x^3 + x^6),\; (1 + x^2)\,(1 - x^2 + x^4 - x^6 + x^8)\}$

Part 4

```
RandomReal[{-1, 1}]
```

-0.360747

```
MyPolynom = Sum[RandomReal[{-1, 1}] x^n, {n, 0, 5}]
```

$-0.579983 - 0.060331\, x + 0.169366\, x^2 + 0.367521\, x^3 - 0.350604\, x^4 + 0.620305\, x^5$

```
Plot[MyPolynom, {x, -5, 5}, ImageSize → 300, Frame → True, PlotRange → {-5, 5}]
```



```
NSolve[MyPolynom == 0, x]
```

```
{{x → -0.64219 - 0.543402 ⅈ}, {x → -0.64219 + 0.543402 ⅈ},
 {x → 0.45474 - 1.09479 ⅈ}, {x → 0.45474 + 1.09479 ⅈ}, {x → 0.94011}}
```

```
NSolve[{MyPolynom == 0 && x ∈ Reals}, x]
```

```
{{x → 0.94011}}
```

```
(* For fun: Plot a few more polynoms *)
SEVERAL = Table[Sum[RandomReal[{-1, 1}] x^n, {n, 0, 5}], {i, 1, 4}];
Plot[SEVERAL, {x, -5, 5}, ImageSize → 300,
 Frame → True, PlotRange → {{-3, 3}, {-5, 5}}, PlotStyle → Thick]
```



Exercise 6: Maximizing the Area of a Rectangle

```
Clear[a, b, AArea, Periphery, asol, amax, bmax];
(* Make sure these are not used *)
```

```
Solve[Periphery == 2 a + 2 b, a] // First
  (* get side a for a given b and Periphery *)
```

$$\left\{a \to \frac{1}{2}\left(-2\,b + \text{Periphery}\right)\right\}$$

```
asol = a /. % (* keep the expression for later *)
```

$$\frac{1}{2}\left(-2\,b + \text{Periphery}\right)$$

```
AArea = a b /. a → asol (* express the area as a function of b and Periphery *)
```

$$\frac{1}{2}\,b\left(-2\,b + \text{Periphery}\right)$$

```
sol = Maximize[AArea, b] (* find b for maximal area *)
```

$$\left\{\frac{\text{Periphery}^2}{16}, \left\{b \to \frac{\text{Periphery}}{4}\right\}\right\}$$

```
bmax = b /. (% // Last) (* remember the solution for b *)
```

$$\frac{\text{Periphery}}{4}$$

```
amax = asol /. b → bmax (* get the solution for a *)
```

$$\frac{\text{Periphery}}{4}$$

```
amax == bmax (* check that it is a square *)
```

True

```
TEST = Periphery → 40;
```

```
Plot[AArea /. TEST, {b, 0, 20}, Frame → True,
 Filling → Axis, GridLines → {{bmax /. TEST}, {}}]
```



Exercise 7: Area of a Circle

```
Clear[x, y, r, sol];
```

```
EQ = x^2 + y^2 == r^2;
```

```
y[r_, x_] = y /. Solve[EQ, y] // Last
```

$\sqrt{r^2 - x^2}$

```
Plot[y[1, x], {x, -1.0, 1.0}, PlotRange → {0, 1},
 AspectRatio → 1 / 2, Filling → Axis, Frame → True]
```




```
$Assumptions = True;
```

```
Integrate[y[r, x], {x, -r, r}] (* this takes very long on my computer *)
```

ConditionalExpression$\left[\frac{\pi r^2}{2}, \text{Re}[r] > 0 \text{ \&\& } \text{Im}[r] == 0\right]$

```
$Assumptions = r > 0; (* Things get simpler with positive r: *)
```

```
Integrate[y[r, x], {x, -r, r}]
```

$\frac{\pi r^2}{2}$

Indefinite Integral

```
sol = Integrate[y[r, x], x] (* general solution *)
```

$\frac{1}{2}\left(x \sqrt{r^2 - x^2} + r^2 \text{ArcTan}\left[\frac{x}{\sqrt{r^2 - x^2}}\right]\right)$

```
sol /. x → 0
```

0

```
Plot[sol /. r → 1, {x, -1, 1}, Frame → True, Filling → Axis]
```




```
sol /. x → r
```

Indeterminate

```
Limit[sol, x → r] (* oups, this gives the wrong sign! *)
```

$-\frac{\pi r^2}{4}$

```
Limit[sol, x → r, Direction → 1] (* we must come from the left! *)
```

$$\frac{\pi\, r^2}{4}$$

Exercise 8: More Plotting

First Part

```
SetOptions[Plot, {Frame → True, Filling → Axis}];
```

```
$Assumptions = {a > 0, λ > 0};
```

```
f[x_] = Sin[a x] Exp[-λ x];
```

```
Reduce[f[1] == 0, a] // Simplify (* Find a such that have a zero at x=1 *)
```

```
C[1] ∈ Integers && (a == 2 π C[1] || π + 2 π C[1] == a)
```

```
% /. C[1] → 0 // Simplify (* pick some constant,
 solution a=0 excluded by $Assumptions *)
```

```
a == π
```

```
Simplify[(f[k] /. a → π) == 0, k ∈ Integers]
(* Check that with a→π, all integer x positions are zeros *)
```

```
True
```

```
Plot[f[x] /. a → π /. λ → {0, 0.5, 1}, {x, 0, 10}]
```



Saw Tooth

```
g[x_, k_] = (-1)^k/k Sin[k x];
```

```
Plot1 = Plot[Table[g[x, k], {k, 1, 4}], {x, -π, 3 π}, ImageSize → 400]
```



```
f[x_, k_] := Sum[g[x, j], {j, 1, k}];
```

```
Plot2 = Plot[f[x, 5], {x, -π, π}, ImageSize → 400]
```



Reverse: Find the Coefficients:

```
Integrate[- x/π Sin[k x], {x, 0, π}]
```

$$\frac{k \pi \text{Cos}[k \pi] - \text{Sin}[k \pi]}{k^2 \pi}$$

```
Simplify[%, k ∈ Integers]
```

$$\frac{(-1)^k}{k}$$

Exercise 9: Drawig a Weel

```
Origin = {0, 0};
```

```
wheel = Table[Line[{Origin, {Cos[2 π phi], Sin[2 π phi]}}], {phi, 0, 1, 1 / 11}];
```

```
Show[Graphics[{Thick, Blue, wheel, Circle[Origin, 1]}],
 Frame → True, ImageSize → 300]
```



Exercise 10: Minimal Interconnect Distance of 4 Points, including Manipulate

The corners are at (0,0), (1,0), ... (1,1)

```
Clear[x, y, l, p0, p1, p2, p3]
```

We assume that we have a ‘H’ shape which is symmetric. Then there is only one free parameter, the distance of the center point at position x (from the left) and 1-x

$$l[x\_] = (1 - 2x) + 4\sqrt{\left(\frac{1}{2}\right)^2 + x^2}\ ;\ (\text{* central part + 4 time the arms *})$$

```
{l[1/2] == 2 √2, l[0] == 3} (* Check tow obvious values *)
```

```
{True, True}
```

```
Minimize[l[x], x]
```

$\{1 + \sqrt{3}, \{x \to \frac{1}{2\sqrt{3}}\}\}$

```
xmin = x /. Last[%];
```

```
xmin // N
```

```
0.288675
```

```
Plot[l[x], {x, 0, 1 / 2}, PlotRange → {2.5, 3}, Frame → True]
```




```
Show[Plot[Evaluate[l[x]], {x, 0, 1 / 2}, PlotRange → {2.5, 3}, Frame → True],
 Graphics[{PointSize[Large], Point[{x, l[x]}] /. x → xmin}]]
(* Add a point to where the minimum is *)
```



```
angle = 2 ArcTan[(1/2)/x /. x → xmin] 360/(2 π) (* Show that all angles are equal *)
```

```
120
```

- **Do some nice Plotting**

Exercise 11: Car lifter

```
SetOptions[Plot, {AspectRatio → 1/GoldenRatio, Frame → True, Filling → Axis}];
```

```
Solve[(h/2)^2 + (x/2)^2 == l^2, h] (* One edge,
half x and half h for a triangle with 90 degrees at the center *)
```

$\{\{h \to -\sqrt{4 l^2 - x^2}\}, \{h \to \sqrt{4 l^2 - x^2}\}\}$

```
h[x_] = h /. Last[%]
```

$\sqrt{4 l^2 - x^2}$

```
Plot[h[x] /. l → 1, {x, 0, 2}]
(* Plot & Check: For x=0, h = 2l, for x=2l, h=0 *)
```




```
-h'[x]
```

$\frac{x}{\sqrt{4 l^2 - x^2}}$

```
Plot[-1/h'[x] /. l → 1, {x, 0, 2}, PlotRange → {0, 100}]
(* h-force * h-distance = constant = x-force * x-dist = const * x
     → hforce = const * dx / dh
*)
```



■ **Make a Graphics of this...**

```
{P1, P2, P3, P4, Pmax} =
  {{0, 0}, {x/2, h[x]/2}, {-x/2, h[x]/2}, {0, h[x]}, {0, h[0]}} /. l → 1;
(* corner points *)
```

```
G = Graphics[{
     Thick
     , Line[{P1, P2}], Line[{P2, P4}],
     Line[{P4, P3}], Line[{P1, P3}] (* four outer lines *)
     , Red, Line[{P2, P3}] (* center line *)
     (*,Point[Pmax]*) (* top point *)
   }, ImageSize → 200];
```

```
Manipulate[
 Show[G /. x → a, PlotRange → {{-1.5, 1}, {0, 2}}],
 {{a, 1, "x"}, 0, 2, Appearance → "Labeled"}
]
```



Exercise 12: Equal Lengths

In[468]:=
```
Clear[x, y, x1, y1, p1, px, K]; (* for safety *)
```

In[469]:=
```
$Assumptions = x > 0 && y > 0 && x1 > 0 && y1 > 0 && K > 0;
```

In[470]:=
```
p1 = {x1, y1}; (* this point is given *)
px = {x, y}; (* unknown point: the kink is here *)
```

In[472]:=
```
L0 = Norm[px]; // Simplify (* distance from origin to unknown (kink) point *)
L1 = Norm[p1 - px]; // Simplify
(* distance from unkown point to given point *)
```

In[474]:=
```
Solve[L0 == L1 && 2 L0 == K, {x, y}] // Last //
 FullSimplify (* equal lengths && sum == K *)
```

Out[474]=
$$\left\{x \to \frac{1}{2}\left(x1 - y1\sqrt{-1 + \frac{K^2}{x1^2 + y1^2}}\right),\ y \to \frac{1}{2}\left(y1 + x1\sqrt{-1 + \frac{K^2}{x1^2 + y1^2}}\right)\right\}$$

In[475]:=
```
pp[x1_, y1_, K_] = {x, y} /. %
```

Out[475]=
$$\left\{\frac{1}{2}\left(x1 - y1\sqrt{-1 + \frac{K^2}{x1^2 + y1^2}}\right),\ \frac{1}{2}\left(y1 + x1\sqrt{-1 + \frac{K^2}{x1^2 + y1^2}}\right)\right\}$$

In[476]:=

```
Manipulate[
 Graphics[Line[{{0, 0}, pp[First[pt], Last[pt], 3], pt}],
  PlotRange → 2, ImageSize → 200]
 , {{pt, {0, 1}}, Locator}
]
```

Out[476]=